

CSE 3302 Notes 4: Semantic Analysis

(Last updated 9/26/12 7:32 AM)

Syntax vs. Semantics

$\Sigma \quad \sigma \quad \Sigma \quad \sigma$

$$\overline{A \vee B} \leftrightarrow \bar{A} \wedge \bar{B}$$

$$(A \rightarrow B) \vee (B \rightarrow A)$$

Well-formed vs. meaningful

Context-free vs. context-sensitive

Static Semantics

Names declared, typechecking

`switch` statement `cases` are independent

Value assigned before use

Dynamic Semantics

What the generated code must do

Language definition . . .

`C` allowed pointer values and loops

4.1. THE ROLE OF THE SEMANTIC ANALYZER

Assertions = Additional properties to be assured at specific execution points

`C/C++` - can be disabled

Dynamic checks - Either built into interpreter or part of generated code

Static analysis

Detects errors

Can avoid some dynamic checking

Can reduce method dispatch costs

4.2. ATTRIBUTE GRAMMARS

General method for defining semantics

Alternatives:

Prototype language implementation (first Ada implementation in SETL, Pascal in Pascal)

Two-level grammar (Algol 68)

Denotational semantics (functional/logic languages)

Natural language . . .

Initially, just a formal method for semantics (Knuth, 1968)

Cornell Program Synthesizer (PL/I, 1981)

Compiler-compilers

E, T, F grammar in book - *synthesized* attributes (bottom-up)

From <http://homepage.cs.uiowa.edu/~slonnegr/plf/Book/>, chapter 3

Strings of form $a^n b^n c^n$ are the only ones acceptable.

Start with grammar:


$\langle \text{letter sequence} \rangle ::= \langle \text{a sequence} \rangle \langle \text{b sequence} \rangle \langle \text{c sequence} \rangle$

$\langle \text{asequence} \rangle ::= \mathbf{a} \mid \langle \text{a sequence} \rangle \mathbf{a}$


$\langle \text{bsequence} \rangle ::= \mathbf{b} \mid \langle \text{bsequence} \rangle \mathbf{b}$

$\langle \text{csequence} \rangle ::= \mathbf{c} \mid \langle \text{csequence} \rangle \mathbf{c}$

Accepted string:



Also accepted (?)



Attribute grammar to capture context-sensitivity with synthesized attributes:

$\langle \text{lettersequence} \rangle ::= \langle \text{asequence} \rangle \langle \text{bsequence} \rangle \langle \text{csequence} \rangle$

condition :

$$\text{Size}(\langle \text{asequence} \rangle) = \text{Size}(\langle \text{bsequence} \rangle) = \text{Size}(\langle \text{csequence} \rangle)$$

$\langle \text{asequence} \rangle ::= \mathbf{a}$



$$\text{Size}(\langle \text{asequence} \rangle) \leftarrow 1$$

| $\langle \text{asequence} \rangle_2 \mathbf{a}$

$$\text{Size}(\langle \text{asequence} \rangle) \leftarrow \text{Size}(\langle \text{asequence} \rangle_2) + 1$$

$\langle \text{bsequence} \rangle ::= \mathbf{b}$
 $\text{Size}(\langle \text{bsequence} \rangle) \leftarrow 1$
 $| \langle \text{bsequence} \rangle_2 \mathbf{b}$
 $\text{Size}(\langle \text{bsequence} \rangle) \leftarrow \text{Size}(\langle \text{bsequence} \rangle_2) + 1$

$\langle \text{csequence} \rangle ::= \mathbf{c}$
 $\text{Size}(\langle \text{csequence} \rangle) \leftarrow 1$
 $| \langle \text{csequence} \rangle_2 \mathbf{c}$
 $\text{Size}(\langle \text{csequence} \rangle) \leftarrow \text{Size}(\langle \text{csequence} \rangle_2) + 1$





Same example, but taking advantage of *inherited* attributes:

```
<lettersequence> ::= <asequence> <bsequence> <csequence>
    InhSize (<bsequence>) ← Size (<asequence>)
    InhSize (<csequence>) ← Size (<asequence>)
```

```
<asequence> ::= a
    Size (<asequence>) ← 1
    | <asequence> 2 a
    Size (<asequence>) ← Size (<asequence> 2) + 1
```

```
<bsequence> ::= b
    condition: InhSize (<bsequence>) = 1
    | <bsequence> 2 b
    InhSize (<bsequence> 2) ← InhSize (<bsequence>) - 1
```

```
<csequence> ::= c
    condition: InhSize (<csequence>) = 1
    | <csequence> 2 c
    InhSize (<csequence> 2) ← InhSize (<csequence>) - 1
```




Meaning of binary numbers using synthesized attributes:

$\langle \text{binary numeral} \rangle ::= \langle \text{binary digits} \rangle_1 \cdot \langle \text{binary digits} \rangle_2$
 $Val(\langle \text{binary numeral} \rangle) \leftarrow Val(\langle \text{binary digits} \rangle_1) +$
 $Val(\langle \text{binary digits} \rangle_2) / 2^{Len(\langle \text{binary digits} \rangle_2)}$

$\langle \text{binary digits} \rangle ::=$
 $\langle \text{binary digits} \rangle_2 \langle \text{bit} \rangle$
 $Val(\langle \text{binary digits} \rangle) \leftarrow 2 \cdot Val(\langle \text{binary digits} \rangle_2) + Val(\langle \text{bit} \rangle)$
 $Len(\langle \text{binary digits} \rangle) \leftarrow Len(\langle \text{binary digits} \rangle_2) + 1$
 | $\langle \text{bit} \rangle$
 $Val(\langle \text{binary digits} \rangle) \leftarrow Val(\langle \text{bit} \rangle)$
 $Len(\langle \text{binary digits} \rangle) \leftarrow 1$

$\langle \text{bit} \rangle ::=$
0
 $Val(\langle \text{bit} \rangle) \leftarrow 0$
 | **1**
 $Val(\langle \text{bit} \rangle) \leftarrow 1$



But only the value of the entire number is computed. What about each bit's contribution?

Nonterminals	Synthesized Attributes	Inherited Attributes
<binary numeral>	<i>Val</i>	—
<binary digits>	<i>Val</i>	<i>Pos</i>
<fraction digits>	<i>Val, Len</i>	—
<bit>	<i>Val</i>	<i>Pos</i>

<binary numeral> ::= <binary digits> . <fraction digits>

$Val(\langle \text{binary numeral} \rangle) \leftarrow Val(\langle \text{binary digits} \rangle) + Val(\langle \text{fraction digits} \rangle)$

$Pos(\langle \text{binary digits} \rangle) \leftarrow 0$

<binary digits> ::=

<binary digits>₂ <bit>

$Val(\langle \text{binary digits} \rangle) \leftarrow Val(\langle \text{binary digits} \rangle_2) + Val(\langle \text{bit} \rangle)$

$Pos(\langle \text{binary digits} \rangle_2) \leftarrow Pos(\langle \text{binary digits} \rangle) + 1$

$Pos(\langle \text{bit} \rangle) \leftarrow Pos(\langle \text{binary digits} \rangle)$

| <bit>

$Val(\langle \text{binary digits} \rangle) \leftarrow Val(\langle \text{bit} \rangle)$

$Pos(\langle \text{bit} \rangle) \leftarrow Pos(\langle \text{binary digits} \rangle)$

<fraction digits> ::=

<fraction digits>₂ <bit>

$Val(\langle \text{fraction digits} \rangle) \leftarrow Val(\langle \text{fraction digits} \rangle_2) + Val(\langle \text{bit} \rangle)$

$Len(\langle \text{fraction digits} \rangle) \leftarrow Len(\langle \text{fraction digits} \rangle_2) + 1$

$Pos(\langle \text{bit} \rangle) \leftarrow - Len(\langle \text{fraction digits} \rangle)$

| <bit>

$Val(\langle \text{fraction digits} \rangle) \leftarrow Val(\langle \text{bit} \rangle)$

$Len(\langle \text{fraction digits} \rangle) \leftarrow 1$

$Pos(\langle \text{bit} \rangle) \leftarrow - 1$


<bit> ::=

0

$Val(\langle \text{bit} \rangle) \leftarrow 0$

| **1**

$Val(\langle \text{bit} \rangle) \leftarrow 2^{Pos(\langle \text{bit} \rangle)}$



Small programming language (Wren) and its context sensitivities

```

program binary is
  var n,p : integer ;
begin
  read n; p := 2;
  while p<=n do p := 2*p end while ;
  p := p/2;
  while p>0 do
    if n>= p then write 1; n := n-p else write 0 end if;
    p := p/2
  end while
end

```

```

<program> ::= program <identifier> is <block>
<block> ::= <declaration seq> begin <command seq> end
<declaration seq> ::= ε | <declaration> <declaration seq>
<declaration> ::= var <variable list> : <type> ;
<type> ::= integer | boolean
<variable list> ::= <variable> | <variable> , <variable list>
<command seq> ::= <command> | <command> ; <command seq>
<command> ::= <variable> := <expr> | skip
  | read <variable> | write <integer expr>
  | while <boolean expr> do <command seq> end while
  | if <boolean expr> then <command seq> end if
  | if <boolean expr> then <command seq> else <command seq> end if

```



```

<expr> ::= <integer expr> | <boolean expr>
<integer expr> ::= <term> | <integer expr> <weak op> <term>
<term> ::= <element> | <term> <strong op> <element>
<element> ::= <numeral> | <variable> | ( <integer expr> ) | - <element>
<boolean expr> ::= <boolean term> | <boolean expr> or <boolean term>
<boolean term> ::= <boolean element>
                    | <boolean term> and <boolean element>
<boolean element> ::= true | false | <variable> | <comparison>
                    | not ( <boolean expr> ) | ( <boolean expr> )
<comparison> ::= <integer expr> <relation> <integer expr>
<variable> ::= <identifier>
<relation> ::= <= | < | = | > | >= | <>
<weak op> ::= + | -
<strong op> ::= * | /
<identifier> ::= <letter> | <identifier> <letter> | <identifier> <digit>
<letter> ::= a | b | c | d | e | f | g | h | i | j | k | l | m
            | n | o | p | q | r | s | t | u | v | w | x | y | z
<numeral> ::= <digit> | <digit> <numeral>
<digit> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

```

Figure 1.8: BNF for Wren

-
1. The program name identifier may not be declared elsewhere in the program.
 2. All identifiers that appear in a block must be declared in that block.
 3. No identifier may be declared more than once in a block.
 4. The identifier on the left side of an assignment command must be declared as a variable, and the expression on the right must be of the same type.
 5. An identifier occurring as an (integer) element must be an integer variable.
 6. An identifier occurring as a Boolean element must be a Boolean variable.
 7. An identifier occurring in a read command must be an integer variable.
-

Figure 1.11: Context Conditions for Wren

```

<reserved word> ::= program | is | begin | end | var | integer
                    | boolean | read | write | skip | while | do | if
                    | then | else | and | or | true | false | not.

```

8. No reserved word may be used as an identifier.

-
1. An attempt is made to divide by zero.
 2. A variable that has not been initialized is accessed.
 3. A **read** command is executed when the input file is empty.
 4. An iteration command (**while**) does not terminate.
-

Figure 1.12: Semantic Errors in Wren

Attribute	Value Types
<i>Type</i>	{ <i>integer</i> , <i>boolean</i> , <i>program</i> , <i>undefined</i> }
<i>Name</i>	String of letters or digits
<i>Var-list</i>	Sequence of Name values
<i>Symbol-table</i>	Set of pairs of the form [Name, Type]

Figure 3.9: Attributes and Values

Nonterminals	Synthesized Attributes	Inherited Attributes
<block>	—	<i>Symbol-table</i>
<declarationsequence>	<i>Symbol-table</i>	—
<declaration>	<i>Symbol-table</i>	—
<variable list>	<i>Var-list</i>	—
<type>	<i>Type</i>	—
<commandsequence>	—	<i>Symbol-table</i>
<command>	—	<i>Symbol-table</i>
<expr>	—	<i>Symbol-table</i> , <i>Type</i>
<integer expr>	—	<i>Symbol-table</i> , <i>Type</i>
<term>	—	<i>Symbol-table</i> , <i>Type</i>
<element>	—	<i>Symbol-table</i> , <i>Type</i>
<boolean expr>	—	<i>Symbol-table</i> , <i>Type</i>
<boolean term>	—	<i>Symbol-table</i> , <i>Type</i>
<boolean element>	—	<i>Symbol-table</i> , <i>Type</i>
<comparison>	—	<i>Symbol-table</i>
<variable>	<i>Name</i>	—
<identifier>	<i>Name</i>	—
<letter>	<i>Name</i>	—
<digit>	<i>Name</i>	—

Figure 3.10: Attributes Associated with Nonterminal Symbols

```

program p is
  var x, y : integer;
  var a : boolean;
begin
  :
end

```




Figure 3.11: Decorated Parse Tree for Wren Program Fragment

Application: Structure Editor

Goal: Editor based on moving among program constructs

Various formatting possibilities - including highlighting, indenting

Use window dimensions in decisions

Eliding of “uninteresting code”

Simple formatting:

```
if (x>y) { temp=x; x=y; y=temp; }
```

```
if (x>y)
{
  temp=x;
  x=y;
  y=temp;
}
```

Synthesized attributes - space needed for construct

Inherited attributes - position of construct in display

Application: Connecting Defs to Uses for Data Flow Analysis

Def: Assigning value to a variable, e.g. (identifier, line#/token#)

Use: Value of variable is used in an expression

Basic block: Straight-line code, no branching

Historic approach: Treat program as flowchart (graph)

Construct-based approach:

INPUT(x) set: defs with some path to construct x without an intervening def

KILL(x) set: defs always invalidated before leaving construct x

GEN(x) set: defs in construct x with path to exit

OUTPUT(x): defs getting through or generated in construct x

$$(\text{INPUT}(x) - \text{KILL}(x)) \cup \text{GEN}(x) = \text{OUTPUT}(x)$$

Each construct (basic block, control structure, function, etc.) has equations and way to match defs to uses.

Could integrate def-use information with structure editor

4.3. EVALUATING ATTRIBUTES

Iterative Firing/Propagation/Queue/Petri Net/Data Flow Machine Ideas

- Build AST

- Initialize obvious values

- Initialize queue with non-finalized tree nodes

- Iterate with updates to queue until changes stop

Integrate With Parser

- OK for static semantics

- Not useful for substantial analysis

- Avoids storage issues

Circularity of attribute grammar

- Does an attribute value at a node depend on itself due to a path of synthesized/inherited attributes?